Appendix J
Three-Phase Reconstruction Code GDTEIT.F

This Fortran 77 program by J. R. Torczynski uses results from the reconstruction codes GDTAXI and EITAXI to determine radial distributions in three-phase, solid-gas-liquid vertical column flows. The algorithm is discussed fully in Section 5.2. An example of the single input file follows.

gdteit_inp.dat:

9.525	column inner radius, R_{col} (cm)
0.0001	gas phase gamma ray attenuation coefficient (1/cm)
0.0856	liquid phase gamma ray attenuation coefficient (1/cm)
0.0866	solid phase gamma ray attenuation coefficient (1/cm)
0.6	Maxwell-Hewitt coefficient, α (unitless)
1.4	$\}C_0$
0.	$\}C_1$ Coefficients of conductivity profile from EITAXI (Eq. 3.8)
0.	$\}C_2$
4	degree n of polynomial fit in GDTAXI reconstruction
0.2	$\}C_0$
0.0	$\}C_2$ Coefficients of phase profile from GDTAXI
0.0	$\}C_4$

The polynomial phase profile from the code GDTAXI takes the form $\varepsilon_L(r) = 1 - \sum_{\substack{i=0 \\ i \text{ even}}}^{n} C_i \left(\frac{r}{R_{col}} \right)^i$.

```
c
c234567890123456789012345678901234567890123456789012345678901234567890123456789012
c
      program gdteit
c
c     Revision 19990420
c
c *** Uses results from gdtaxi and eitaxi
c     to determine three-phase material distribution.
c
      implicit double precision (a-h,o-z)
```

```
c
      parameter (ngdtm=10)
      parameter (neitm=2)
      parameter (nr=50)
c
      dimension cgdt(0:ngdtm)
      dimension ceit(0:neitm)
      dimension eg(0:nr), el(0:nr), es(0:nr)
      dimension radnrm(0:nr)
      dimension raddim(0:nr)
      dimension cndnrm(0:nr)
      dimension gamnrm(0:nr)
c
 1001 format (1x,d18.12)
 1002 format (1x,i4)
 1003 format (1x,i4,1x,d18.12)
 1004 format (7(1x,d11.5))
 2000 format (1x,a)
 2001 format (1x,a12,d18.12)
 2002 format (1x,a12,i4)
c
c *** Read physical parameters and profile coefficients.
c
      write (6,2000) 'Reading input parameters from gdteit_inp.dat'
      open (unit=21, status='old', file='gdteit_inp.dat')
      read (21,*) radius
      read (21,*) gamgas
      read (21,*) gamliq
      read (21,*) gamsol
      read (21,*) heweta
      neit = neitm
      do 0100 ieit = 0, neit, 1
         read (21,*) ceit(ieit)
 0100    continue
      read (21,*) ngdt2
      ngdt = ngdt2 / 2
      do 0150 igdt = 0, ngdt, 1
         read (21,*) cgdt(igdt)
 0150    continue
      close (unit=21)
      write (6,2001) '   radius = ', radius
      write (6,2001) '   gamgas = ', gamgas
      write (6,2001) '   gamliq = ', gamliq
      write (6,2001) '   gamsol = ', gamsol
      write (6,2001) '   heweta = ', heweta
      do 0200 ieit = 0, neit, 1
         write (6,2001) '   ceit(i)= ',  ceit(ieit)
 0200    continue
      write (6,2002) '   ngdt2  = ', ngdt2
      do 0250 igdt = 0, ngdt, 1
         write (6,2001) '   cgdt(i)= ',  cgdt(igdt)
 0250    continue
c
c *** Determine the gas, liquid, and solid volume fraction profiles.
c
      write (6,2000) 'Computing results'
      gampar = ((gamsol - gamgas) / (gamliq - gamgas)) - 1.
      do 0400 ir = 0, nr, 1
         rn = dfloat(ir) / dfloat(nr)
         radnrm(ir) = rn
         raddim(ir) = rn * radius
         cn = 1.
         cn = cn + ceit(1) * (2. * rn ** 2 - 1.)
```

```
         cn = cn + ceit(2) * (1. - 6. * rn ** 2 + 6. * rn ** 4)
         cn = cn / ceit(0)
         cndnrm(ir) = cn
         gn = cgdt(0)
         do 0300 igdt = 1, ngdt, 1
            gn = gn + cgdt(igdt) * rn ** (2 * igdt)
 0300       continue
         gamnrm(ir) = gn
         call phase3(heweta,cn,gampar,gn,eg0,el0,es0)
         eg(ir) = eg0
         el(ir) = el0
         es(ir) = es0
 0400    continue
c
c *** Determine the average volume fractions.
c
      avgeg  = radnrm(0) * eg(0) + radnrm(nr) * eg(nr)
      avgel  = radnrm(0) * el(0) + radnrm(nr) * el(nr)
      avges  = radnrm(0) * es(0) + radnrm(nr) * es(nr)
      avgcnd = radnrm(0) * cndnrm(0) + radnrm(nr) * cndnrm(nr)
      avggam = radnrm(0) * gamnrm(0) + radnrm(nr) * gamnrm(nr)
      do 0500 ir = 1, nr-1, 2
         avgeg  = avgeg  + 4. * radnrm(ir) * eg(ir)
         avgel  = avgel  + 4. * radnrm(ir) * el(ir)
         avges  = avges  + 4. * radnrm(ir) * es(ir)
         avgcnd = avgcnd + 4. * radnrm(ir) * cndnrm(ir)
         avggam = avggam + 4. * radnrm(ir) * gamnrm(ir)
 0500    continue
      do 0550 ir = 2, nr-2, 2
         avgeg  = avgeg  + 2. * radnrm(ir) * eg(ir)
         avgel  = avgel  + 2. * radnrm(ir) * el(ir)
         avges  = avges  + 2. * radnrm(ir) * es(ir)
         avgcnd = avgcnd + 2. * radnrm(ir) * cndnrm(ir)
         avggam = avggam + 2. * radnrm(ir) * gamnrm(ir)
 0550    continue
      avgeg  = avgeg  * 2. / ( 3. * dfloat(nr) )
      avgel  = avgel  * 2. / ( 3. * dfloat(nr) )
      avges  = avges  * 2. / ( 3. * dfloat(nr) )
      avgcnd = avgcnd * 2. / ( 3. * dfloat(nr) )
      avggam = avggam * 2. / ( 3. * dfloat(nr) )
      write (6,2001) '   avgeg  = ', avgeg
      write (6,2001) '   avgel  = ', avgel
      write (6,2001) '   avges  = ', avges
      write (6,2001) '   avgcnd = ', avgcnd
      write (6,2001) '   avggam = ', avggam
c
c *** Write profile file.
c
      write (6,2000) 'Writing profiles to gdteit_gls.dat'
      open (unit=28, status='unknown', file='gdteit_gls.dat')
      do 0700 ir = nr, 1, -1
         write (28,1004) -radnrm(ir), -raddim(ir),
     1      eg(ir), el(ir), es(ir), cndnrm(ir), gamnrm(ir)
 0700    continue
      do 0750 ir = 0, nr, 1
         write (28,1004) radnrm(ir), raddim(ir),
     1      eg(ir), el(ir), es(ir), cndnrm(ir), gamnrm(ir)
 0750    continue
c
c *** Write output file.
c
      write (6,2000) 'Writing output parameters to gdteit_out.dat'
      open (unit=29, status='unknown', file='gdteit_out.dat')
```

```
      write (29,1001) radius
      write (29,1001) gamgas
      write (29,1001) gamliq
      write (29,1001) gamsol
      write (29,1001) heweta
      do 0800 ieit = 0, neit, 1
         write (29,1001) ceit(ieit)
 0800    continue
      write (29,1002) ngdt2
      do 0850 igdt = 0, ngdt, 1
         write (29,1001) cgdt(igdt)
 0850    continue
      write (29,1001) avgeg
      write (29,1001) avgel
      write (29,1001) avges
      write (29,1001) avgcnd
      write (29,1001) avggam
      close (unit=29)
c
c *** Stop, end.
c
      stop 'gdteit'
      end
c
c234567890123456789012345678901234567890123456789012345678901234567890123456789012
c
      subroutine phase3(heweta,cn,gampar,gn,eg0,el0,es0)
c
c *** Solves three simultaneous equations
c     (phases sum to unity, gdt, eit) analytically
c     for gas, liquid, solid volume fractions.
c     Uses modified Maxwell-Hewitt relation: (1-x)/(1+a*x)
c     Uses this relation recursively for solid-liquid, gas-solliqmix.
c
      implicit double precision (a-h,o-z)
c
      a = heweta
      b = gn
      c = gampar
      d = cn
c
      aa = 1. + c - a * a * d + a * c * d
      bb = a * a * b * d + c * d - (1. + b + 2. * c + a * d + a * c * d)
      cc = b + c + a * b * d - c * d
c
      eg0 = - ( bb + sqrt(bb * bb - 4. * aa * cc) ) / ( 2. * aa )
      tmp = ( 1. - eg0 ) / ( 1. + a * eg0 )
      tmp = d / tmp
      tmp = ( 1. - tmp ) / ( 1. + a * tmp )
      es0 = ( 1. - eg0 ) * tmp
      el0 = 1. - eg0 - es0
c
      return
      end
c
c234567890123456789012345678901234567890123456789012345678901234567890123456789012
c
```